

VSTTE'10 Competition Q1 in ProofPower-HOL (v2)

Rob Arthan

18th August 2010

1 Introduction

Isn't anybody in the VSTTE world interested in functional programming? To address the balance, here is an attempt based on primitive recursive functions in **ProofPower**-HOL executed by rewriting.

This document is a literate script containing the specs, "code" and proof commands. See the theory listing towards the end of the document for a collected summary of the definitions and results.

2 Red Tape

SML

```
| open_theory "hol";  
| new_theory "vstte10-1";  
| set_merge_pcs ["hol1"];
```

3 The Spec

We need some specs (albeit pretty close to code).

HOL Constant

	<i>NatSum</i> : $\mathbb{N} \text{ LIST} \rightarrow \mathbb{N}$
<hr/>	
	$\text{NatSum } [] = 0$
\wedge	$\forall n \text{ ns} \bullet \text{NatSum } (\text{Cons } n \text{ ns}) = n + \text{NatSum } ns$

HOL Constant

	<i>NatMax</i> : $\mathbb{N} \text{ LIST} \rightarrow \mathbb{N}$
<hr/>	
	$\text{NatMax } [] = 0$
\wedge	$\forall n \text{ ns} \bullet \text{NatMax } (\text{Cons } n \text{ ns}) = \text{if } \text{NatMax } ns < n \text{ then } n \text{ else } \text{NatMax } ns$

4 The Code

Now the code, implemented using the fold function from the library as any functional programmer would (cursing the designer for the order of the last two arguments):

$\mathbf{SumMax} : \mathbb{N} \text{ LIST} \rightarrow \mathbb{N} \times \mathbb{N}$
$\forall ns \bullet$
$\mathbf{SumMax} \ ns =$
$\mathbf{Fold} (\lambda n \ (s, m) \bullet (n + s, \text{if } m < n \text{ then } n \text{ else } m)) \ ns \ (0, 0)$

5 Testing, Testing, ...

ML bindings for the defining theorems:

SML

```
val nat_sum_def = get_spec  $\lceil \mathbf{NatSum} \rceil$ ;
val nat_max_def = get_spec  $\lceil \mathbf{NatMax} \rceil$ ;
val sum_max_def = get_spec  $\lceil \mathbf{SumMax} \rceil$ ;
```

See the theory listing towards the end of the document for the results of the following tests.

First test the specs (Sum is 40, my eye!):

SML

```
val nat_sum_test_thm = save_thm ("nat_sum_test_thm",
  rewrite_conv [nat_sum_def]
   $\lceil \mathbf{NatSum} \ [9;5;0;2;7;3;2;1;10;6] \rceil$ );
val nat_max_test_thm = save_thm ("nat_max_test_thm",
  rewrite_conv [nat_max_def]
   $\lceil \mathbf{NatMax} \ [9;5;0;2;7;3;2;1;10;6] \rceil$ );
```

Now test the code:

SML

```
val sum_max_test_thm = save_thm ("sum_max_test_thm",
  rewrite_conv [sum_max_def, fold_def]
   $\lceil \mathbf{SumMax} \ [9;5;0;2;7;3;2;1;10;6] \rceil$ );
```

6 Correctness Proof

SML

```
set_goal([],  $\lceil$ 
   $\mathbf{SumMax} \ [] = (0, 0) \wedge$ 
 $\forall n \ ns \bullet \mathbf{SumMax} \ (\mathbf{Cons} \ n \ ns) =$ 
   $(n + \mathbf{Fst} \ (\mathbf{SumMax} \ ns),$ 
   $\text{if } \mathbf{Snd}(\mathbf{SumMax} \ ns) < n \text{ then } n \text{ else } \mathbf{Snd}(\mathbf{SumMax} \ ns))$ 
 $\rceil$ );
a(rewrite_tac [sum_max_def, fold_def]);
val sum_max_prim_rec_thm = save_pop_thm "sum_max_prim_rec_thm";
```

SML

```

| set_goal([],  $\ulcorner$ 
|  $\forall ns \bullet \text{SumMax } ns = (\text{NatSum } ns, \text{NatMax } ns)$ 
|  $\urcorner$ );
| a(strip_tac);
| a(list_induction_tac  $\ulcorner ns \urcorner$  THEN
|   asm_rewrite_tac[sum_max_prim_rec_thm, nat_sum_def, nat_max_def]);
| val sum_max_correct_thm = save_pop_thm "sum_max_correct_thm";

```

7 Property Proof

SML

```

| set_goal([],  $\ulcorner$ 
|  $\forall ns \bullet \text{NatSum } ns \leq \text{Length } ns * \text{NatMax } ns$ 
|  $\urcorner$ );
| a(strip_tac);
| a(list_induction_tac  $\ulcorner ns \urcorner$  THEN
|   asm_rewrite_tac[length_def, nat_sum_def, nat_max_def]);
| a(REPEAT strip_tac THEN cases_tac  $\ulcorner \text{NatMax } ns < x \urcorner$ 
|   THEN asm_rewrite_tac[]);
| (* *** Goal "1" *** *)
| a(POP_ASM_T ante_tac THEN
|   rewrite_tac[rewrite_rule[ $\leq$ _def]
|   (pc_rule1 "lin_arith" prove_rule[]
|      $\ulcorner \forall a \ b : \mathbb{N} \bullet a < b \Leftrightarrow a + 1 \leq b \urcorner$ )]
|   THEN REPEAT strip_tac
|   THEN all_var_elim_asm_tac1);
| a(PC_T1 "lin_arith" asm_prove_tac[]);
| (* *** Goal "2" *** *)
| a(POP_ASM_T ante_tac THEN
|   rewrite_tac[rewrite_rule[ $\leq$ _def]
|   (pc_rule1 "lin_arith" prove_rule[]
|      $\ulcorner \forall a \ b : \mathbb{N} \bullet \neg a < b \Leftrightarrow b \leq a \urcorner$ )]
|   THEN REPEAT strip_tac);
| a(PC_T1 "lin_arith" asm_prove_tac[]);
| val nat_sum_nat_max_estimate_thm = save_pop_thm "nat_sum_nat_max_estimate_thm";

```

8 THE THEORY vstte10-1

8.1 Parents

hol

8.2 Constants

$NatSum \quad \mathbb{N} \text{ LIST} \rightarrow \mathbb{N}$
 $NatMax \quad \mathbb{N} \text{ LIST} \rightarrow \mathbb{N}$
 $SumMax \quad \mathbb{N} \text{ LIST} \rightarrow \mathbb{N} \times \mathbb{N}$

8.3 Definitions

$NatSum \quad \vdash NatSum [] = 0$
 $\quad \quad \quad \wedge (\forall n \ ns \bullet NatSum (Cons n \ ns) = n + NatSum \ ns)$
 $NatMax \quad \vdash NatMax [] = 0$
 $\quad \quad \quad \wedge (\forall n \ ns$
 $\quad \quad \quad \bullet NatMax (Cons n \ ns)$
 $\quad \quad \quad = (if \ NatMax \ ns < n \ then \ n \ else \ NatMax \ ns))$
 $SumMax \quad \vdash \forall \ ns$
 $\quad \quad \bullet SumMax \ ns$
 $\quad \quad = Fold$
 $\quad \quad (\lambda n \ (s, m) \bullet (n + s, (if \ m < n \ then \ n \ else \ m)))$
 $\quad \quad ns$
 $\quad \quad (0, 0)$

8.4 Theorems

$nat_sum_test_thm$
 $\quad \vdash NatSum [9; 5; 0; 2; 7; 3; 2; 1; 10; 6] = 45$
 $nat_max_test_thm$
 $\quad \vdash NatMax [9; 5; 0; 2; 7; 3; 2; 1; 10; 6] = 10$
 $sum_max_test_thm$
 $\quad \vdash SumMax [9; 5; 0; 2; 7; 3; 2; 1; 10; 6] = (45, 10)$
 $sum_max_prim_rec_thm$
 $\quad \vdash SumMax [] = (0, 0)$
 $\quad \quad \wedge (\forall n \ ns$
 $\quad \quad \bullet SumMax (Cons n \ ns)$
 $\quad \quad = (n + Fst (SumMax \ ns),$
 $\quad \quad (if \ Snd (SumMax \ ns) < n$
 $\quad \quad then \ n$
 $\quad \quad else \ Snd (SumMax \ ns))))$
 $sum_max_correct_thm$
 $\quad \vdash \forall \ ns \bullet SumMax \ ns = (NatSum \ ns, NatMax \ ns)$
 $nat_sum_nat_max_estimate_thm$
 $\quad \vdash \forall \ ns \bullet NatSum \ ns \leq Length \ ns * NatMax \ ns$

9 INDEX

<i>NatMax</i>	1
<i>NatMax</i>	4
<i>NatSum</i>	1
<i>NatSum</i>	4
<i>nat_max_def</i>	2
<i>nat_max_test_thm</i>	2
<i>nat_max_test_thm</i>	4
<i>nat_sum_def</i>	2
<i>nat_sum_nat_max_estimate_thm</i>	3
<i>nat_sum_nat_max_estimate_thm</i>	4
<i>nat_sum_test_thm</i>	2
<i>nat_sum_test_thm</i>	4
<i>SumMax</i>	2
<i>SumMax</i>	4
<i>sum_max_correct_thm</i>	3
<i>sum_max_correct_thm</i>	4
<i>sum_max_def</i>	2
<i>sum_max_prim_rec_thm</i>	2
<i>sum_max_prim_rec_thm</i>	4
<i>sum_max_test_thm</i>	2
<i>sum_max_test_thm</i>	4